YU. A. GRIGORYEV, A. V. TSIGANOV
V. A. Fock Institute of Physics,
St. Petersburg State University,
198504, St. Petersburg, Russia
E-mails: tsiganov@mph.phys.spbu.ru; com974@mail.ru

SYMBOLIC SOFTWARE FOR SEPARATION OF VARIABLES IN THE HAMILTON–JACOBI EQUATION FOR THE L-SYSTEMS

Received May 10, 2005; accepted July 30, 2005 DOI: 10.1070/RD2005v010n04ABEH000323

We discuss a computer implementation of the known algorithm of finding separation coordinates for a special class of orthogonal separable systems called L-systems or Benenti systems.

Contents

1. Introduction

Let $\mathcal{Q}$ be a n-dimensional Riemannian manifold with generic local coordinates $q = (q^1, q^2, \dots, q^n)$ and positive-definite metric tensor $\mathbf{G}$.

On the cotangent bundle $T^*\mathcal{Q}$ of $\mathcal{Q}$ with canonical coordinates (p, q) we consider dynamical system with the natural Hamilton function

$$H = T(p,q) + V(q) = \frac{1}{2}\sum_{i,j=1}^{n} \mathrm{g}^{ij}(q)p_i p_j + V(q). \tag{1.1}$$

Here $\mathrm{g}^{ij}(q)$ are the components of the metric tensor $\mathbf{G}$ and $V(q)$ is the potential energy, a smooth function on $\mathcal{Q}$ canonically lifted to a function on $T^*\mathcal{Q}$.

One of the most effective ways to solve the corresponding equations of motion is to apply separation of variables in the Hamilton–Jacobi equation

$$H(p,q) = E. \tag{1.2}$$

A coordinate system $Q = (Q^1, \dots, Q^n)$ is called separable if the Hamilton–Jacobi equation admits a complete solution of the form

$$\mathcal{S}(Q,\alpha) = \sum_{i=1}^{n} \mathcal{S}_i(Q^i,\alpha), \qquad \det\left[\frac{\partial^2 \mathcal{S}}{\partial Q^i \partial \alpha^j}\right] \neq 0\,.$$

Here $\alpha = (\alpha^1, \dots, \alpha^n)$ is a set of separation constants. The corresponding Jacobi equations

$$P_i \equiv \partial_i \mathcal{S} = \frac{\partial \mathcal{S}_i(Q^i,\alpha)}{\partial Q^i}$$

Mathematics Subject Classification: 37J35, 37K10, 70H20
Key words and phrases: integrable systems, Hamilton-Jacobi equation, separation of variables

are called the separated equations. In a similar manner, we will call a natural Hamiltonian or a potential *separable* if such a separable coordinate system exists.

For a given Hamilton function $H(p,q)$ the problem of finding canonical transformation from initial variables (p,q) to separation coordinates (P,Q) is very non-trivial. The problem was originally formulated by Jacobi when he invented elliptic coordinates and successfully applied them to solve several important mechanical problems, such as the problem of geodesic motion on an ellipsoid, the Euler problem of planar motion in a force field of two attracting centers and the problem of the motion of three particles, which interact due to forces depending on their relative distances [1].

Up to now the search of separation coordinates for a given integrable system has been more of an art than a science. However, for a special class of the natural Hamiltonians we have a complete and algorithmic solution of this problem [4], [5], [6], [7]. The algorithm is sufficiently straightforward to be implemented on a computer and thus has turned into a practical tool.

In this note we present an implementation of this algorithm by means of the computer algebra system Maple 9.5.

2. Algorithm of the point separation of variables

According to [4], [6] we confine ourselves to the search of a point canonical transformations $Q = f(q)$ and $P = g(q,p)$ only.

In this case from Stäckel (1893), Levi-Civita (1904), Eisenhart (1934), Kalnins & Miller (1980) and Benenti (1993) we have

Proposition 1. *The Hamilton–Jacobi equation* (1.2) *is separable in orthogonal coordinates if and only if there exists a Killing* 2*-tensor* $\mathbf{K}$ *with simple eigenvalues and normal eigenvectors such that*

$$d(\mathbf{K}dV) = 0, \tag{2.1}$$

where d *denotes the exterior derivative.*

Such a tensor $\mathbf{K}$ is called *characteristic* and its existence is completely determined by the kinetic energy T. It means that the equation $T+V=E$ is separable only if the geodesic equation is separable. The equation $d(\mathbf{K}dV)=0$ is an integrability condition for the existence of the potential $V(q)$ that may be added to T.

In 1992 Benenti suggested a simple recurrence procedure to construct a special family of Killing tensors $\mathbf{K}$ obeying the assumptions of this theorem. He considered a special class of Riemannian manifolds $\mathcal{Q}$ endowed with the L-tensor.

Following Benenti let us call *an L-tensor* a conformal Killing tensor $\mathbf{L}$ with vanishing torsion and pointwise simple eigenvalues Q_i. Under these conditions the tensors

$$\mathbf{K}_j = \sum_{k=0}^{j} (-1)^k \sigma_{j-k} \mathbf{L}^k, \qquad \text{or} \qquad \mathbf{K}_j = \sigma_j \mathbf{G} - \mathbf{K}_{j-1}\mathbf{L}\,, \tag{2.2}$$

with σ_a being the elementary symmetric polynomials of degree a in the eigenvalues of $\mathbf{L}$ are the Killing tensors with simple eigenvalues and normal eigenvectors.

As noticed in [8], a tensor $\mathbf{L}$ is an L-tensor with respect to the usual Riemannian metric $\mathbf{G}$ if

$$d(\mathcal{L}_{X_T}\,\theta - T d\sigma_1) = 0, \tag{2.3}$$

where $\mathcal{L}$ is the Lie derivative along the geodesic vector field X_T, $\sigma_1 = \operatorname{tr}\mathbf{L}$ is a symmetric polynomial and

$$\theta = \sum_{i,j=1}^{n} L^i_j\, p_i dq^j \tag{2.4}$$

is the L-deformation of the standard Liouville 1-form $\theta_0 = \sum p_j dq^j$ for any set of fibered coordinates (p,q).

The equation $d(\mathbf{K}dV) = 0$ with the Killing tensor $\mathbf{K}_1$ (2.2) may be rewritten in the similar form

$$d(\mathcal{L}_{X_V}\,\theta - V d\sigma_1) = 0\,, \tag{2.5}$$

see for instance [10].

Below we will use the following well-known expression for the Lie derivative $\mathcal{L}$ along the vector field X

$$\mathcal{L}_X = \mathtt{i}_X\, d + d\,\mathtt{i}_X\,.$$

Here $\mathtt{i}_X$ is a hook operator and d is an exterior derivative. Since $d^2 = 0$, the amount of intermediate calculations can be reduced and we get

$$d\mathcal{L}_X = d\,\mathtt{i}_X\,d + d^2\,\mathtt{i}_X = d\,\mathtt{i}_X\,d.$$

In this notation equations (2.3) and (2.5) read as

$$d(\mathtt{i}_{X_T} d\,\theta - T d\sigma_1) = 0, \tag{2.6}$$
$$d(\mathtt{i}_{X_V} d\,\theta - V d\sigma_1) = 0\,. \tag{2.7}$$

We will call any separable orthogonal system whose Killing tensor $\mathbf{K}$ in (2.1) is generated by L-tensor according to (2.2) *an L-system* or *a Benenti system*. To construct the separation coordinates Q for the L-system we have to solve equations (2.6) and (2.7) for the functions $L^i_j(q)$ and find the eigenvalues of the tensor $\mathbf{L}$.

In the next section we present a computer program for search of L-tensors and the associated separation coordinates for L-systems.

Remark 1. The 1-form θ may be used to construct the second Poisson structure on $T^*\mathcal{Q}$ [9], [10]. In this case the recursion operator N will be a complete lifting of the L-tensor $\mathbf{L}$ from the configuration space to the whole phase space. Therefore, the coordinate separation of variables can be treated as a particular case of the bi-Hamiltonian theory of separation of variables.

3. The program for calculation of separation variables

In this section we realize the algorithm discussed above using the symbolic computational system Maple v.9.5.

We will use the standard Maple package `liesymm`, which was designed for construction of differential forms from partial differential equations. In fact, we are doing the inverse procedure, i. e. starting with the differential forms (2.6)–(2.7) we have to get a system of partial differential equations.

Let us start with the following command

```
> with(liesymm):
```

to be able to use short names to access the procedures of this Maple package at the interactive level.

As the first step we need to determine the dimension of a given configuration space $\mathcal{Q}$ and to assume that the phase space $T^*\mathcal{Q}$ is equipped with some canonical coordinates $q = (q^1, \ldots, q^n)$ and $p = (p_1, \ldots, p_n)$:

```
> n := 2;
```

$$n := 2$$

```
> q:=seq(q||i,i=1..n): p:=seq(p||i,i=1..n): var:=q,p:
> setup(var);
```

$$[q1,\ q2,\ p1,\ p2]$$

Now we can define the first symmetric polynomial σ_1 on the eigenvalues of $\mathbf{L}$ and L-deformation θ (2.4) of the standard Liouville form

```
> sigma:=add(L[i,i](q),i=1..n);
```

$$\sigma := L_{1,1}(q1,\, q2) + L_{2,2}(q1,\, q2)$$

```
> theta:=add(add(L[i,j](q)*p||i*d(q||j),i=1..n),j=1..n);
```

$$\theta := L_{1,1}(q1,\, q2)\, p1\, \mathrm{d}(q1) + L_{2,1}(q1,\, q2)\, p2\, \mathrm{d}(q1) + \\ + L_{1,2}(q1,\, q2)\, p1\, \mathrm{d}(q2) + L_{2,2}(q1,\, q2)\, p2\, \mathrm{d}(q2)$$

Here the components of the L-tensor $L^i_j(q)$ are unknown functions on the configuration space $\mathcal{Q}$.

At the second step we introduce the canonical Poisson tensor $\mathscr{P}$

```
> ed:=array(identity, 1..n,1..n): u:=array(sparse,1..n,1..n):
> P:=linalg[stackmatrix](linalg[augment](u,ed),
                         linalg[augment](-ed,u));
```

$$P := \begin{bmatrix} 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \\ -1 & 0 & 0 & 0 \\ 0 & -1 & 0 & 0 \end{bmatrix}$$

and calculate the vector fields $X_T = \mathscr{P}dT(p,q)$ and $X_V = \mathscr{P}dV(q)$, that is,

```
> dT:=array(1..2*n):
> for i from 1 to 2*n do dT[i]:=diff(T(var),var[i]): end do:
> X_T:=evalm(P&*dT);
```

$$XT := \Big[\frac{\partial}{\partial p1}\,\mathrm{T}(q1,\, q2,\, p1,\, p2),\, \frac{\partial}{\partial p2}\,\mathrm{T}(q1,\, q2,\, p1,\, p2) - \\ - (\frac{\partial}{\partial q1}\,\mathrm{T}(q1,\, q2,\, p1,\, p2)),\, -(\frac{\partial}{\partial q2}\,\mathrm{T}(q1,\, q2,\, p1,\, p2))\Big]$$

and

```
> dV:=array(1..2*n):
> for i from 1 to 2*n do dV[i]:=diff(V(q),var[i]): end do:
> X_V:=evalm(P&*dV);
```

$$XV := \Big[0,\, 0,\, -(\frac{\partial}{\partial q1}\,\mathrm{V}(q1,\, q2)),\, -(\frac{\partial}{\partial q2}\,\mathrm{V}(q1,\, q2))\Big]$$

Since some components of the vector field X_V are zero, the corresponding equations are simpler than those for X_T. Therefore, we start with the second equation (2.7) in order to explain some features of the Maple procedures. Namely, substituting the vector field X_V and 1-form θ into the $\mathtt{i}_{X_V}\, d\theta$ **,** one gets

```
> idT:=wcollect(hook(d(theta),X_V)):
```

$$idT := \Big(- L_{2,1}(q1,\, q2)\,(\frac{\partial}{\partial q2}\,\mathrm{V}(q1,\, q2))(q1,\, q2,\, p1,\, p2) - \\ - L_{1,1}(q1,\, q2)\,(\frac{\partial}{\partial q1}\,\mathrm{V}(q1,\, q2))(q1,\, q2,\, p1,\, p2)\Big)\mathrm{d}(q1) + \\ + \Big(- L_{2,2}(q1,\, q2)\,(\frac{\partial}{\partial q2}\,\mathrm{V}(q1,\, q2))(q1,\, q2,\, p1,\, p2) - \\ - L_{1,2}(q1,\, q2)\,(\frac{\partial}{\partial q1}\,\mathrm{V}(q1,\, q2))(q1,\, q2,\, p1,\, p2)\Big)\mathrm{d}(q2)$$

It is easy to see that this expression contains derivatives of the potential $V(q)$, which formally depend on all of the variables (q,p). In fact these derivatives depend on q only. It is an unpleasant feature of the Maple procedure `hook` from the package `liesymm`.

In order to get the correct expression for $\mathtt{i}_{X_V}\, d\theta$ we have to use the special substitution

```
> Trans:={seq(diff(V(q),q||i)(var)=diff(V(q),q||i),i=1..n)}:
> idT:=subs(Trans,idT);
```

$$idT := \Big(- L_{2,1}(q1,\, q2)\,\frac{\partial}{\partial q2}\,\mathrm{V}(q1,\, q2) - L_{1,1}(q1,\, q2)\,\frac{\partial}{\partial q1}\,\mathrm{V}(q1,\, q2)\Big)\mathrm{d}(q1) +$$
$$+\Big(- L_{2,2}(q1,\, q2)\,\frac{\partial}{\partial q2}\,\mathrm{V}(q1,\, q2) - L_{1,2}(q1,\, q2)\,\frac{\partial}{\partial q1}\,\mathrm{V}(q1,\, q2)\Big)\mathrm{d}(q2)$$

Equation (2.7) is satisfied identically and the coefficients for the independent 2-forms in (2.7) vanish. It gives rise to the following system of equations

```
> SysEq:=annul(d(idT-V(q)*d(sigma)), [var] ): nops(SysEq);
```

At $n = 2$ equation (2.7) generates just one partial differential equation, while at $n = 3$ the system `SysEq` consists of three equations and so on.

We keep these equations in a special list

```
> ListEq:=NULL:
> for i from 1 to nops(SysEq) do
>     ListEq:=ListEq,lhs(SysEq[i]);
> end do:
```

Now we proceed with the remaining equation (2.6). As above we have to simplify this equation through the additional substitution for derivatives after the application of the `hook` operator

```
> Eq:=wcollect(d(hook(d(theta),X_T)-T(var)*d(sigma))):
> Trans:={seq( diff(T(var),var[i])(var)=diff(T(var),var[i]),i=1..2*n)}:
```

Expanding equation (2.6) in the basis of 2-forms one gets one more system of equations

```
> SysEq:=annul(subs(Trans,Eq),[var]): nops(SysEq);
```

At $n = 2, 3$ equation (2.7) gives rise to six and fifteen equations respectively.

By adding these equations to the list of equations `ListEq` one gets a complete set of algebraic and partial differential equations on the components of L-tensor.

```
> for i from 1 to nops(SysEq) do ListEq:=ListEq,lhs(SysEq[i]); end do:
> NumEq:=nops([ListEq]);
```

$$NumEq := 7$$

At $n = 2, 3$ one gets seven and eighteen equations in four and nine functions $L^i_j(q)$ respectively.

At the third step we have to define the Hamilton function. As an example we consider the Hénon-Heiles integrable model with the Hamiltonian

$$H = p_1^2 + p_2^2 + \frac{1}{2}(q_1^2 + q_2^2) + q_1^2 q_2 + 2q_2^3\,. \tag{3.1}$$

In this case the kinetic energy T and potential $V(q)$ are given by

```
> Tn:=add(p||i^2,i=1..n);
> Vn:=1/2*(q1^2+q2^2)+q1^2*q2+2*q2^3;
```

Substituting these parts of the Hamilton function into the prepared above list of equations `ListEq` one gets a set of polynomial equations of the second degree in the momenta p, which must be identically satisfied for all admissible values of variables p. It means that coefficients of the second, first and zeroth power of p_i vanish. All these coefficients form a new system of algebraic and partial differential equations in the functions $L^i_j(q)$.

```
> System:= NULL:
> for i from 1 to NumEq do
>     u:=expand(dvalue(subs({ T(var)=Tn, V(z)=Vn },ListEq[i]))):
>     System:=System,coeffs(u,{p}):
> end do:
> NumSys:=nops({System});\vspace{-1em}
```

$$NumSys := 13$$

At $n = 2, 3, 4$ the resulting system of equation consists of 13, 51 and 136 equations. Even though some of these equations may be dependent, it is convenient to use all of them simultaneously. Of course, for a generic potential this system is heavily overdetermined and has only the trivial solution, which means that the Hamilton function is non-separable.

In the case of polynomial potentials we can easily solve this system by means of the standard Maple procedure `pdsolve`

```
> Ans:=pdsolve( {System}, {seq(seq(L[i,j](z), i=1..n),j=1..n)} );
```

which allows us to get the following answer

$$Ans := \{L_{1,1}(q1,\, q2) = \frac{3\,C_1}{4} + C_2,\, L_{1,2}(q1,\, q2) = \frac{C_1\, q1}{2},$$
$$L_{2,2}(q1,\, q2) = C_1\, q2 + C_2,\, L_{2,1}(q1,\, q2) = \frac{C_1\, q1}{2}\}$$

depending on two arbitrary constants $C_{1,2}$.

Substituting this answer into the L-tensor one gets

```
> Trans:={seq(seq(L[i,j](q)=L[i,j],i=1..n),j=1..n)}:
> Ans:=simplify(map2(subs,Trans,Ans)):
> L:=array(1..n,1..n):
> L:=map2(subs,Ans,evalm(L));
```

$$L := \begin{bmatrix} \frac{3}{4}\,C_1 + C_2 & \frac{C_1\, q1}{2} \\ \frac{C_1\, q1}{2} & C1\ q2 + C_2 \end{bmatrix}$$

The eigenvalues of this L-tensor are the separation coordinates

```
> e:=collect(det(L-ed*lambda),lambda,factor):
> Q1:=solve(e,lambda)[1]; Q2:=solve(e,lambda)[2];
```

$$Q1 := \frac{3C_1}{8} + C_2 + \frac{C_1\, q2}{2} + \frac{\sqrt{9\,C_1^2 - 24\,C_1^2\, q2 + 16\,C_1^2\, q2^2 + 16\,C_1^2\, q1^2}}{8},$$
$$Q2 := \frac{3\,C_1}{8} + C_2 + \frac{C_1\, q2}{2} - \frac{\sqrt{9\,C_1^2 - 24\,C_1^2\, q2 + 16\,C_1^2\, q2^2 + 16\,C_1^2\, q1^2}}{8}..$$

Thus one gets translated parabolic coordinates, which are the separation coordinates for the Hénon-Heiles model [7]. It takes less than a minute of the computer time.

More explicitly, we obtain a family of equivalent separable coordinate systems labeled by $C_{1,2}$. Recall, that two separable systems are called equivalent if the corresponding separated solutions of the Hamilton–Jacobi equation generate the same Lagrangian foliation of $T^*\mathcal{Q}$.

REMARK 2. Using a standard personal computer, for $n \leqslant 10$ and for the polynomial potentials we can solve the overdetermined system of equations `System` with the standard procedure `pdsolve` in a reasonable time. For $n > 10$ we have to use special computers or special packages for solving heavily overdetermined systems of equations.

4. Examples

In this section we present some examples which illustrate various aspects of the proposed code.

The anharmonic oscillator. The configuration space $\mathcal{Q}$ is the n-dimensional Euclidean space $\mathbb{R}^n$ with Cartesian coordinates q and metric $ds^2 = \sum g_{ij} dq_i dq_j = \sum {dq_i}^2$. Let us consider the anharmonic oscillator with the Hamiltonian

$$H = \sum_{i=1}^{n} p_i^2 + \sum_{i=1}^{n} a_i q_i^2 + \left(\sum_{i=1}^{n} q_i^2 \right)^2 .$$

We have to substitute this Hamiltonian for (3.1) using the commands

```
> Tn:=add(p||i^2,i=1..n);
> Vn:=add(a||i*q||i^2,i=1..n)+(add(q||i^2,i=1..n))^2;
```

It is easy to calculate the solution of the complete system of equations `System` obtained from equations (2.6) and (2.7)

$$L_j^i(q) = (C_1 + a_i - a_n)\delta_{ij} + C_2\, q_i q_j$$

Here the solution found by Maple is presented without any further simplification.

For $C_1 = a_n$ and $C_2 = 1$ the eigenvalues Q_i of the corresponding tensor $\mathbf{L}$ are defined by the equation

$$\frac{\det(L-\lambda)}{\prod_{i=1}^{n}(\lambda - a_i)} \equiv -1 + \sum_{i=1}^{n} \frac{{q_i}^2}{\lambda - a_i} = \prod_{i=1}^{n} \frac{\lambda - Q_i}{\lambda - a_i} . \tag{4.1}$$

This well-known relation defines standard elliptic coordinates in $\mathbb{R}^n$.

These coordinates were introduced by Jacobi in a note in Crelle's Journal [11]. A thorough discussion of their general properties as well as of their use for separation of variables in the Hamilton–Jacobi equation can be found in his lecture notes [1].

The Euler problem. Let us consider the Euler problem of planar motion in a force field of two attracting centers with the Hamiltonian

$$H = p_1^2 + p_2^2 - \left(\frac{a_1}{\sqrt{(q_1 - c)^2 + q_2^2}} + \frac{a_2}{\sqrt{(q_1 + c)^2 + q_2^2}} \right) .$$

In contrast with the previous examples the potential $V(q)$ is not a polynomial but an algebraic function.

The solution of the overdetermined system of equations obtained from (2.6) and (2.7) is as follows:

$$L = \begin{bmatrix} (q_1^2 - c^2)C_2 + C_1 & q_1 q_2 C_2 \\ q_1 q_2 C_2 & q_2^2 C_2 + C_1 \end{bmatrix}$$

For $C_1 = 1/2\, c^2$ and $C_2 = 1$ the eigenvalues of the matrix L coincide with the standard elliptic coordinates

$$1 - \frac{q_1^2}{\lambda + c^2/2} - \frac{q_2^2}{\lambda - c^2/2} = \frac{(\lambda - Q_1)(\lambda - Q_2)}{(\lambda - c^2/2)(\lambda + c^2/2)}$$

The Toda lattice. As above, the configuration space $\mathcal{Q}$ is the n-dimensional Euclidean space $\mathbb{R}^n$ with Cartesian coordinates q. The Hamiltonian of the periodic Toda lattice is given by

$$H = \frac{1}{2} \sum_{i=1}^{n} {p_i}^2 + \sum_{i=1}^{n} \exp(q_i - q_{i+1}) , \qquad q_{n+1} \equiv q_1 . \tag{4.2}$$

For $n = 2$ the unique solution of the complete system of equations `System` is

$$L = \begin{bmatrix} C_1 & C_2 \\ C_2 & C_1 \end{bmatrix}$$

Of course, the eigenvalues of this matrix are not the separation variables. However, in this case the number of free parameters is equal to the dimension of the Riemannian manifold $\mathcal{Q}$. It allows us to construct the separation coordinates (see [7]). Namely, let us diagonalize the matrix L

$$L = V^{-1}\mathrm{diag}(C_1 + C_2, C_1 - C_1)V, \qquad V = \begin{pmatrix} 1 & 1 \\ -1 & 1 \end{pmatrix}.$$

It is easy to see that the separation coordinates Q are the following Cartesian coordinates

$$Q = Vq, \qquad \Rightarrow \qquad Q_1 = q_1 + q_2, \quad Q_2 = q_1 + q_2\,.$$

At $n = 3$ solution of the equations `System` is trivial

$$L = \begin{bmatrix} C_1 & C_2 & C_2 \\ C_2 & C_1 & C_2 \\ C_2 & C_2 & C_1 \end{bmatrix}$$

The number of free parameters is less than the dimension of the Riemannian manifold. It means that this Hamiltonian is not separable through point transformations. However, we have to underline that this Hamiltonian admits separation of variables in a wider class of canonical transformations of the whole phase space.

The Neumann system. The unit sphere $\mathbb{S}^n$ is the Riemannian subspace of $\mathbb{R}^{n+1}$ whose points have Cartesian coordinates $x = (x_1, \ldots, x_{n+1})$ satisfying $|x| = \sqrt{(x, x)} = 1$.

The Neumann system is a well-known and profoundly studied mechanical system, which describes the motion of a particle on a sphere under the influence of a quadratic potential $V = \frac{1}{2}\sum a_i x_i^2$. Due to the form of the constraint and the potential, it is quite natural to perform all the computations using the local coordinates $q_i = x_i^2$, $i = 1, \ldots, n$. In these coordinates the Hamiltonian of the Neumann system is given by $H = T + V$ where

$$T = 2\sum_{i=1}^{n} q_i(1 - q_i)p_i^2 - 4\sum_{i<j}^{n} q_i q_j p_i p_j, \qquad V = \frac{1}{2}\sum_{i=1}^{n}(a_i - a_{n+1})q_i\,.$$

Here p_i are momenta conjugate to q_i and $a_1, \ldots, a_{n+1}$ are arbitrary parameters [10]. We can insert this Hamiltonian into the Maple code with the commands

```
> Tn:=2*add(q||i*(1-q||i)*p||i^2,i=1..n)
>    -4*add(add(q||i*q||j*p||i*p||j,j=i+1..n),i=1..n);
> Vn:=1/2*add((a||i-a||(N+1))*q||i,i=1..n);
```

For $n = 2$ the unique solution of the complete system of equations `System` looks like

$$L = \begin{bmatrix} C_2 - \dfrac{q_1(a_3 - a_1) + a_1 - a_2}{a_2 - a_3}\, C_1 & q_1\, C_1 \\ \dfrac{q_2\,(a_1 - a_3)}{a_2 - a_3}\, C_1 & q_2\, C_1 + C_2 \end{bmatrix}$$

Recall, that $L^i_j(q)$ are the components of the tensor $\mathbf{L}$, which is symmetric with respect to the metric $\mathbf{G}$. For the non-flat metrics the L-matrices obtained in Maple are non-symmetric.

If $C_1 = a_3 - a_2$ and $C_2 = a_2$ the eigenvalues $Q_{1,2}$ of this L-tensor are zeroes of the following function

$$e(\lambda) = \frac{\det(L-\lambda)}{\prod_{i=1}^{n}(\lambda - a_i)} = \frac{q_1}{\lambda - a_1} + \frac{q_2}{\lambda - a_2} + \frac{1 - q_1 - q_2}{\lambda - a_3}$$

In terms of the redundant coordinates x_i the separation coordinates read

$$\sum_{j=1}^{n+1} \frac{x_j^2}{\lambda - a_j} = \frac{\prod_{i=1}^{n}(\lambda - Q_i)}{\prod_{j=1}^{n+1}(\lambda - a_j)}, \qquad \text{with} \qquad \sum_{i=1}^{n+1} x_j^2 = 1\,.$$

These relations are known to define the spheroconical or elliptic-spherical coordinates. For $n > 2$ one gets the same relations.

For $n = 2, 3$ all the computation takes about one and eight minutes respectively.

Remark 3. In contrast with [10] we directly solve equations (2.3)–(2.5) without any additional assumptions about the affine structure of the solutions. It allows us to prove the uniqueness of a solution.

The Jacobi-Calogero inverse-square model. The configuration space $\mathcal{Q}$ is 3-dimensional Euclidean space $\mathbb{R}^3$ with Cartesian coordinates $q = (q_1, q_2, q_3)$. The Hamilton function is given by

$$H = \sum_{i=1}^{3} p_i^2 + (q_1 - q_2)^{-2} + (q_2 - q_3)^{-2} + (q_3 - q_1)^{-2}\,.$$

For this Hamiltonian the solution of the complete system of equations `System` depends on four arbitrary parameters

$$L = \begin{bmatrix} q_1^2 C_1 + 2q_1 C_2 + C_3 & q_1 q_2 C_1 + (q_1 + q_2) C_2 + C_4 & q_1 q_3 C_1 + (q_1 + q_3) C_2 + C_4 \\ q_1 q_2 C_1 + (q_1 + q_2) C_2 + C_4 & q_2^2 C_1 + 2q_2 C_2 + C_3 & q_2 q_3 C_1 + (q_2 + q_3) C_2 + C_4 \\ q_1 q_3 C_1 + (q_1 + q_3) C_2 + C_4 & q_2 q_3 C_1 + (q_2 + q_3) C_2 + C_4 & q_3^2 C_1 + 2q_3 C_2 + C_3 \end{bmatrix}$$

The number of free parameters exceeds the dimension of the Riemannian manifold. It means that this integrable system is a degenerate or superintegrable system, which is separable in some different coordinate systems. Namely, at the different values of the parameters $C_1, \ldots, C_4$ the eigenvalues of the tensor $\mathbf{L}$ are oblate spheroidal, prolate spheroidal, spherical, rotational parabolic and circular cylindrical coordinates. The detailed discussion may be found in [7], [12].

An elliptic egg. Let us consider the Hamiltonian with rational potential

$$H = \sum_{i=1}^{n} p_i^2 + \frac{c}{\left(1 - \sum_{i=1}^{n} \frac{q_i^2}{a_i^2}\right)}, \qquad c, a_i \in \mathbb{R}\,.$$

In contrast with the previous examples with polynomial potentials, the potential $V(q)$ is a rational function. This potential has a singular surface, which is an ellipsoid

$$\sum_{i=1}^{n} \frac{q_i^2}{a_i^2} = 1.$$

For $c > 0$, this singular surface is repelling inwards so that a trajectory initiated within the ellipsoid will remain there forever.

For $n > 2$ the Maple procedure `pdsolve` cannot solve the system of equations `System` obtained from equations (2.6) and (2.7) in a reasonable time.

In this case we have to solve the equation for the geodesic motion (2.6) and then the equation in potential (2.7). For $n \leqslant 10$ it takes just a few minutes.

In our case the solution of the system of equations generated by the geodesic equation (2.6) is

$$L^i_j(q) = \alpha q_i q_j + A_{ij} q_i + B_{ij} q_j + C_{ij} \,.$$

Here the coefficients α, A_{ij}, B_{ij} and C_{ij} are arbitrary constants, $A_{ij} = B_{ji}$ and $C_{ij} = C_{ji}$.

Substituting this solution into the system of equations obtained from (2.7) one gets a system of algebraic equations in the coefficients A_{ij}, B_{ij} and C_{ij}. The standard Maple program `solve` easily solves this system of algebraic equations in few seconds. The eigenvalues of the corresponding tensor $\mathbf{L}$ are the standard elliptic coordinates (4.1).

5. Conclusion

We present the first part of the symbolic software which builds separation coordinates for the Hamilton–Jacobi equation of the L-systems using methods from the differential geometry. The second part of this software is intended to build and solve the corresponding separated equations. We leave this part to subsequent publications.

References

[1] *C. G. J. Jacobi.* Vorlesungen über Dynamik. Georg Reimer, Berlin. 1866. Jacobi's lectures on dynamics given in Königsberg 1842–1843, published by A.Clebsch

[2] *P. Stäckel.* Uber die Integration der Hamilton–Jacobischen Differential Gleichung Mittelst Separation der Variabel. Habilitationsschrift, Halle. 1891.

[3] *T. Levi-Civita.* Integrazione delle equazione di Hamilton-Jacobi per separazione di variabili. Math. Ann. 1904. V. 24. P. 383–397.

[4] *L. P. Eisenhart.* Separable systems of Stäckel. Math. Ann. 1934. V. 35. P. 284–305.

[5] *E. G. Kalnins, W. Miller, Jr.* Killing tensors and variable separation for Hamilton–Jacobi and Helmholtz equations. SIAM J. Math. Anal. 1980. V. 11. P. 1011–1026.

[6] *S. Benenti.* Orthogonal separable dynamical systems. In: *O. Kowalski, D. Krupka (Eds.)* Proceedings of the 5th International Conference on Differential Geometry and Its Applications, Silesian University at Opava, August 24–28, 1992. Diff. Geom. Appl. 1993. V. 1. P. 163–184.

[7] *S. Rauch-Wojciechowski, C. Waksjö.* How to find separation coordinates for the Hamilton–Jacobi equation: a criterion of separability for natural Hamiltonian systems. Math. Phys. Anal. Geom. 2003. V. 6. №4. P. 301–348.

[8] *M. Crampin, W. Sarlet, G. Thompson.* Bi-differential calculi, bi-Hamiltonian systems and conformal Killing tensors. J. Phys. A. 2000. V. 33. P. 8755–8770. *M. Crampin.* Conformal Killing tensors with vanishing torsion and the separation of variables in the Hamilton–Jacobi equation. Diff. Geom. Appl. 2003. V. 18. P. 87-102.

[9] *A. Ibort, F. Magri, G. Marmo.* Bihamiltonian structures and Stäckel separability. J. Geometry and Physics. 2000. V. 33. P. 210–228.

[10] *C. Bartocci, G. Falqui, M. Pedroni.* A geometric approach to the separability of the Neumann–Rosochatius system. Diff. Geom. Appl. 2004. V. 21. №3. P. 349–360.

[11] *C. G. J. Jacobi.* J. Reine Angew. Math. XIX.

[12] *S. Benenti, C. Chanu, G. Rastelli.* The super-separability of the three-body inverse-square Calogero system. J. Math. Phys. 2000. V. 41. P. 4654–4678.